#include <RBDdimmer.h>
#include <OneWire.h>
#include <DallasTemperature.h>

// -------- MAX6675 --------
const int thermoSO1 = 8;
const int thermoCS1 = 9;
const int thermoSCK1 = 10;

const int thermoSO2 = 6;
const int thermoCS2 = 7;
const int thermoSCK2 = 10;

// -------- Dimmer --------
const int pinDimmer = 5;
const int pinZC = 2;           // (no usado por RBDdimmer en este sketch, pero reservado)
dimmerLamp dimmer(pinDimmer);

// -------- PID Variables --------
float Setpoint = 75;
float Kp = 7.0, Ki = 0.8, Kd = 1.2;
float input = 0, output = 0, error = 0;
float integral = 0, previous_error = 0;
unsigned long previousMillis = 0;
const long interval = 1000;

// -------- Modo de funcionamiento --------
String modo = "horizontal";  // "horizontal" o el que uses

// -------- DS18B20 en pin 3 --------
#define ONE_WIRE_BUS 3
OneWire oneWire(ONE_WIRE_BUS);
DallasTemperature ds18b20(&oneWire);

// -------- Funciones --------
double readMAX6675(int CS, int SCK, int SO) {
  digitalWrite(CS, LOW);
  delayMicroseconds(10);
  uint16_t value = 0;
  for (int i = 15; i >= 0; i--) {
    digitalWrite(SCK, LOW);
    delayMicroseconds(1);
    if (digitalRead(SO)) {
      value |= (1 << i);
    }
    digitalWrite(SCK, HIGH);
    delayMicroseconds(1);
  }
  digitalWrite(CS, HIGH);
  if (value & 0x4) return -1.0;  // error de sonda abierta
  value >>= 3;
  return value * 0.25;
}

void setup() {
  Serial.begin(9600);
  delay(2000);

  // MAX6675
  pinMode(thermoSO1, INPUT);
  pinMode(thermoCS1, OUTPUT);
  pinMode(thermoSCK1, OUTPUT);
  digitalWrite(thermoCS1, HIGH);

  pinMode(thermoSO2, INPUT);
  pinMode(thermoCS2, OUTPUT);
  pinMode(thermoSCK2, OUTPUT);
  digitalWrite(thermoCS2, HIGH);

  // Dimmer
  dimmer.begin(NORMAL_MODE, ON);
  dimmer.setPower(0);

  // DS18B20
  ds18b20.begin();

  // PLX-DAQ etiquetas
  Serial.println("CLEARDATA");
  Serial.println("LABEL,DATE,TIME,TempControl(C),TempMonitoreo(C),Setpoint,Potencia(%),Modo,DS18B20(C)");
}

void loop() {
  unsigned long currentMillis = millis();
  if (currentMillis - previousMillis >= interval) {
    previousMillis = currentMillis;

    // Sensores MAX6675
    double tempControl = readMAX6675(thermoCS1, thermoSCK1, thermoSO1);
    double tempMonitoreo = readMAX6675(thermoCS2, thermoSCK2, thermoSO2);

    // PID según Setpoint
    if (Setpoint <= 55) {
      Kp = 7;   Ki = 0.8; Kd = 1.3;
    } else if (Setpoint > 55 && Setpoint <= 70) {
      Kp = 7.5; Ki = 0.9; Kd = 1.5;
    } else {
      Kp = 8;   Ki = 1.1; Kd = 1.7;
    }

    float deltaT = (modo == "horizontal") ? (Setpoint - tempControl) : (Setpoint - tempMonitoreo);

    if (deltaT <= 15 && deltaT > 1) {
      output = map(deltaT, 15, 2, 70, 40);
      integral = 0;
    } else if (deltaT <= 1) {
      error = deltaT;
      integral += error * (interval / 1000.0);
      float derivative = (error - previous_error) / (interval / 1000.0);
      output = Kp * error + Ki * integral + Kd * derivative;
      previous_error = error;
    } else {
      output = 95;
      integral = 0;
    }

    if (output < 0)  output = 0;
    if (output > 95) output = 95;

    dimmer.setPower(output);

    // -------- DS18B20 --------
    ds18b20.requestTemperatures();
    float tempDS = ds18b20.getTempCByIndex(0);
    // Manejo de errores típicos: -127 (desconectado), 85 (lectura por defecto)
    bool dsOk = !(tempDS == DEVICE_DISCONNECTED_C || tempDS == -127.00 || tempDS == 85.00);

    // -------- PLX-DAQ --------
    Serial.print("DATA,DATE,TIME,");
    Serial.print((tempControl < 0) ? "NaN," : String(tempControl) + ",");
    Serial.print((tempMonitoreo < 0) ? "NaN," : String(tempMonitoreo) + ",");
    Serial.print(Setpoint); Serial.print(",");
    Serial.print(output);   Serial.print(",");
    Serial.print(modo);     Serial.print(",");
    if (dsOk) {
      Serial.println(tempDS);
    } else {
      Serial.println("NaN");
    }
  }
}

